/**
 * Copyright (c) 2012, Ben Fortuna
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions
 * are met:
 *
 *  o Redistributions of source code must retain the above copyright
 * notice, this list of conditions and the following disclaimer.
 *
 *  o Redistributions in binary form must reproduce the above copyright
 * notice, this list of conditions and the following disclaimer in the
 * documentation and/or other materials provided with the distribution.
 *
 *  o Neither the name of Ben Fortuna nor the names of any other contributors
 * may be used to endorse or promote products derived from this software
 * without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
 * "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
 * LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR
 * A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR
 * CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL,
 * EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO,
 * PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR
 * PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF
 * LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING
 * NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
 * SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
 */
package net.fortuna.ical4j.model


import spock.lang.Specification

class ModelSerializationSpec extends Specification {
	
	def 'test object serialization'() {
		setup: 'serialise object'
		File ser = File.createTempFile('CalendarSerializationSpec', '.dat')
		ser.withObjectOutputStream { oout ->
			oout << modelObject
		}

		and: 'de-serialise object'
		def deserialised
		ser.withObjectInputStream { oin ->
			deserialised = oin.readObject()
		}

		expect: 'deserialised object equals test object'
		assert deserialised == modelObject
		
		where:
		modelObject << [
			new Calendar(),
		
        	new ContentBuilder().calendar {
	            prodid '-//Ben Fortuna//iCal4j 1.0//EN'
	            version '2.0'
	            vevent {
	                uid '1'
	                dtstamp()
	                dtstart('20090810', parameters: parameters { value 'DATE' })
	                action 'DISPLAY'
	                attach('http://example.com/attachment', parameters: parameters() { value 'URI' })
	            }
	        },
		
			new DateList()
		]
	}
}
